



Paper CT09

SAS outputs in Excel workbook using ODS Excel

Meghana Mahajani, Cytel Statistical Software & Services Pvt. Ltd., Pune, India

ABSTRACT

Excel is a common tool across various industries. It has many functions which answer statistical, engineering and financial questions. In the clinical trials domain, Word is the preferred tool for generation of reports for cross-function sharing. However, considering Excel has its own benefits over Word while handling the data, by using functions like filter and PIVOT tables, we received a request to generate patient profiles with different subgroups in Excel. After careful consideration, we chose ODS Excel to generate the patient profiles in Excel. This presentation will share experience on ODS Excel, its different options with examples, and will highlight its effectiveness over the conventional method of PROC EXPORT.

INTRODUCTION

Many times, Clinical research scientist and safety reviewer wants SAS data in Excel format to explore data for exploratory analysis or addressing health authority queries. Our team received a request to generate patient profiles in Excel workbook for different subgroups. When we started working on this request, there were some questions, how to write SAS data to Excel workbook? Can we write SAS data in multiple worksheets of Excel workbook? Is it possible to get fully customized outputs? Can we highlight any specific value? We found that there are multiple ways to write SAS dataset into Excel workbook such as ODS TAGSETS.ExcelXP, PROC Export, ODS Excel. Among these, we chose ODS Excel which writes SAS data to Excel easily and customized output as we like using ODS Excel options and different levels of options. ODS Excel can write output generated from any SAS procedures such as PROC PRINT, PROC REPORT, PROC FREQ, PROC MEANS.

In this paper, we will cover examples that use PROC PRINT and PROC REPORT features, ODS Excel options which are used for,

- Highlight the cell by any colour
- Create multiple worksheets for different subgroups
- Wrap column headers
- Add sheet names and colour to sheet
- Apply freeze panes
- Add filters to columns header
- Title and footnotes
- Apply different styles for layout

Although the ODS Excel supports Graphics procedures, we will not cover these procedures in this paper. Also, Instead of Patient Profiles, we will discuss subject level data in this paper.

Let's look at a SAS dataset ADSL which will be used to discuss ODS Excel options and outputs generated by ODS Excel.

Dataset ADSL, source for the examples in this paper.

	Subject	Treatment	Gender	Agecat	Height_Cm	Weight_Kg	Bmi	Score
1	Subject1	3	F	> 45 Yrs	168.9	57.6	20.191248986	2
2	Subject2	1	F	> 45 Yrs	168.9	132.9	46.587100526	3
3	Subject3	2	F	>25 to <=35 Yrs	177.8	109.3	34.574558945	3
4	Subject4	2	F	>35 to <=45 Yrs	163.8	113.4	42.265426881	3
5	Subject5	1	M	>25 to <=35 Yrs	168.9	63.5	22.259449838	2.5
6	Subject6	2	F	> 45 Yrs	170.2	64.9	22.404001099	2
7	Subject7	1	F	> 45 Yrs	180.3	95.3	29.315779549	2.5
8	Subject8	1	F	>25 to <=35 Yrs	175.3	77.1	25.089415618	2
9	Subject9	3	F	<= 25 Yrs	165	77.1	28.319559229	2.5
10	Subject10	1	M	>35 to <=45 Yrs	183	77.1	23.022484995	1
11	Subject11	3	M	>25 to <=35 Yrs	177.8	122	38.591913919	1
12	Subject12	3	M	>25 to <=35 Yrs	170.2	70.8	24.440728472	2

SAS DATA TO EXCEL WORKBOOK

Before going to customized output generation by options of ODS Excel, let us first discuss about general syntax of ODS Excel that writes SAS data to Excel workbook with the help of an example. SAS code 1 which will print SAS data ADSL in Excel workbook using SAS procedures PROC Print. This SAS code write SAS data ADSL in Excel file only for variables subject, score, treatment, gender mentioned in var statement of PROC Print.

SAS code 1 - SAS data ADSL in the Excel workbook

```
ods excel file = "<Path>\filename.xlsx"; /* Start named Excel file */
proc print data=ADSL;
  var subject score treatment gender; /* Any SAS procedure in between ods
  run;                                excel file ="" and ods excel close */
ods excel close; /* Stop writing to Excel file */
```

Output 1 - Output generated by SAS Code 1; SAS data ADSL in Excel workbook

	A	B	C	D	E
1	Obs	Subject	Score	Treatment	Gender
2	1	Subject1	2.0	3	F
3	2	Subject2	3.0	1	F
4	3	Subject3	3.0	2	F
5	4	Subject4	3.0	2	F
6	5	Subject5	2.5	1	M
7	6	Subject6	2.0	2	F
8	7	Subject7	2.5	1	F
9	8	Subject8	2.0	1	F
10	9	Subject9	2.5	3	F
11	10	Subject10	1.0	1	M

Print 1 - Data Set WORK.TEMP

CREATING MULTIPLE WORKSHEETS

Now, let us see how we can create multiple worksheets for different subgroups. SAS dataset ADSL has a variable 'Agecat' with four different categories. Multiple worksheets were created in Excel workbook for these four Age categories. See the SAS code in Appendix SAS code 2 which will print SAS data ADSL in Excel workbook using PROC report. The by agecat statement reported output in four different Excel worksheets

Output 2 - Output generated by appendix SAS code 2; SAS data ADSL in multiple worksheets for four different age categories in an Excel workbook

	A	B	C	D	E	F	G	H	I	J
1	Subject ID	Treatment	Gender	Height (in cm)	weight (in kg)	BMI	Score			
2	Subject9		3 F	155	77.1	28.319559	2.5			
3	Subject14		1 M	177.8	124.3	39.319466	2			
4	Subject16		1 F	157.5	59	23.784329	1			
5	Subject19		3 F	165.1	46.9	17.205952	2			
6	Subject22		2 F	167.6	82.6	29.405734	3			
7	Subject31		3 F	167.6	63.1	22.463702	1.5			
8	Subject36		1 M	172.7	83	27.828732	2			
9	Subject42		1 M	167.6	69.4	24.706512	2			
10	Subject44		1 F	170.2	92.1	31.793659	2.5			
11	Subject53		3 F	175.3	91.6	29.807918	2.5			
12	Subject54		1 F	162.6	102.5	38.768838	1.5			
13	Subject55		3 M	172.7	104.3	34.970322	2.5			
14	Subject73		2 F	170.2	62.6	21.610023	1			
15	Subject74		2 F	163.8	63.1	23.518064	2.5			
16	Subject77		2 F	172.7	53.5	17.937797	2.5			
17	Subject82		2 M	172.7	83.5	27.996375	2.5			
18	Subject95		3 F	170.2	101.6	35.073136	2.5			
19	Subject97		1 F	156.2	55.7	22.829328	1			
20										
21										
22										
23										
24										
25										
26										
27										

Report 1 - Detailed and-or s | Report 2 - Detailed and-or s | Report 3 - Detailed and-or s | Report 4 - Detailed and-or s



CUSTOMIZING OUTPUT USING ODS EXCEL OPTIONS

Once SAS data or output is in Excel workbook, it can be customized as we like by using suboptions of the options, style, title. See SAS code *Appendix SAS code 3* which created customized output 3 using ODS Excel and with the help of PROC Report features.

Format minscore applied at define statement of 'Score' highlight score cell by red which are between 0-1. Similar way, if statistician required, we can highlight the p-values which are significant or non-significant with different colours for different data. `style(header)=[tagattr="wrap:yes"]` mentioned at proc report defined statement wraps the column headers.

Suboptions of OPTIONS

There are multiple suboptions of Options of ODS Excel destination. We used few below options to customize our output as shown in Output 3.

Suboptions with value	Possible values	Description
<code>embedded_titles = "on"</code>	on off	Sets whether title(s) are embedded in the spreadsheet
<code>embed_footnotes_once= "on"</code>	on off	Sets whether footnote(s) are embedded in the spreadsheet
<code>autofilter = "1-7"</code>	all range	Add filters to column(s)
<code>tab_color = "Red"</code>	Colour	Apply specified colour to tab
<code>frozen_headers = "on"</code>	on off number	Freeze Panes at column headers and fixed the row
<code>frozen_rowheaders = "2"</code>	on of numer	Freeze Panes at row headers and fixed the column
<code>start_at = "2,2"</code>	column,row	Start writing table from specified column and row
<code>absolute_column_width = "15,16,13,10,15,15,14,14"</code>	Number	Apply specified column width
<code>sheet_name = "#byval1"</code>	any name	Assign specified name as sheet name #byval1 gives tab name as first variable named in the BY statement used in PROC REPORT
<code>row_heights= "30,20,50,20,20,25,5"</code>	Number	Apply row height for rows that contained table header, table body, by value lines, titles, footers, page break sequentially.

ODS STYLE

Now, we are ready with our customized Excel workbook with multiple worksheets with required titles and footnotes. What will be the option if someone wants to change output layout? Yes, we can change the layout of our Excel output. `style=` option permits to change the look of output. For output 3, we used `style=Listing` to change default layout to listing layout. There are different styles available for ODS to format output.

Available ODS styles

Analysis	Barrettsblue	Blockprint	Dtree	Listing	Printer
Daisy	Default	Dove	Egdefault	monochromeprinter	SASdocprinter
Excel	Fancyprinter	Festival	festivalprinter	nofontdefault	SASweb
Gantt	Grayscaleprinter	Htmlblue	Harvest	Pearl	normalprinter
Highcontrast	Highcontrastlarge	Journal	journal1a	powerpointlight	Plateau
journal2	journal2a	journal3	journal3a	Sapphire	Raven
Normal	Meadow	Meadowprinter	Minimal	seasideprinter	Ocean
Pearlj	Monospace	moonflower	Netdraw	Vadark	powerpointdark
Valight	Word	Seaside	statistical	Statdoc	Rtf
Vahighcontrast					

Output 3 - Output generated by appendix SAS code 3; Customized report in Excel workbook in multiple worksheets for four different subgroups

	A	B	C	D	E	F	G	H
1								
2		Study ID: AABB						
3		Listing for subjects with baseline information						
4		Age Category: <= 25 Yrs						
6		Baseline Information						
7		Subject ID	Treatment	Gender	Subject Height (in cm)	Subject Weight (in kg)	Subject BMI	Score
23		Subject82	2	M	172.7	83.5	27.996375	2.5
24		Subject95	3	F	170.2	101.6	35.073136	2.5
25		Subject97	1	F	156.2	55.7	22.829328	1
27		Note: Subjects listed in the output who received at least one Study treatment.						
28		Note: Subjects listed in the output whose baseline age is <= 25 Yrs						
30								
31								
32								
33								
34								
35								
36								
37								
38								

SUMMARY TABLE IN SAME EXCEL WORKBOOK

Until now, this paper described how we can write and customize SAS data in Excel workbook in multiple worksheets using a SAS Procedure. Also, using with multiple SAS procedures, we can add multiple SAS outputs in Excel workbook in different worksheets or a single worksheet. See SAS code *Appendix SAS code 4* created output 4 which reported summary dataset 'all_stat'. Output 4 is an Excel workbook with worksheet 'Summary' added as compared to output 3. Two different PROC reports called under the same ODS Excel File="Filename" with different ODS Excel option and suboptions.

Output 4 - Output generated by appendix SAS code 4; Additional worksheet 'summary' in output 3

	A	B	C	D	E	F	G	H	I	
1										
2		Summary table for Baseline Characteristic								
4		Age category	Treatment	Sex	Treatment N	Sex N	Height Mean(SD)	Weight Mean(SD)	BMI Mean(SD)	Score Mean(SD)
5		<= 25 Yrs	1	F	4	4	161.63 (6.349)	77.33 (23.491)	29.29 (7.487)	1.50 (0.707)
6	M			3	3	172.70 (5.100)	92.23 (28.591)	30.62 (7.695)	2.00 (0.000)	
7	2		F	4	4	168.58 (3.804)	65.45 (12.255)	23.12 (4.789)	2.25 (0.866)	
8			M	1	1	172.70 (NA)	83.50 (NA)	28.00 (NA)	2.50 (NA)	
9	3	F	5	5	168.64 (4.291)	76.06 (21.870)	26.57 (6.899)	2.20 (0.447)		
10		M	1	1	172.70 (NA)	104.30 (NA)	34.97 (NA)	2.50 (NA)		
12	>25 to <=35 Yrs	1	F	10	10	169.68 (7.838)	83.24 (23.115)	28.88 (7.748)	1.90 (1.125)	
13			M	7	7	176.53 (6.278)	82.59 (14.572)	26.64 (5.773)	1.86 (0.556)	
14		2	F	10	10	170.44 (7.519)	90.64 (18.981)	31.36 (7.089)	1.85 (0.914)	
15			M	4	4	187.33 (7.594)	109.30 (5.581)	31.22 (2.254)	1.38 (0.750)	
16		3	F	6	6	166.13 (3.991)	72.33 (11.596)	26.35 (5.118)	1.42 (1.114)	
17			M	5	5	174.24 (7.314)	90.10 (21.986)	29.45 (5.694)	1.60 (0.894)	
18										

ODS EXCEL EFFECTIVENESS OVER PROC EXPORT AND LIMITATIONS

Using ODS Excel, we can generate customized reports or graphs as we like, whereas PROC Export can export SAS data into Excel workbook, but it couldn't generate customized reports or generate graphs. If SAS data is large with lots of variables and observations, ODS Excel doesn't perform well. In this case PROC Export is preferable. Also, PROC Export is useful when updates are in certain worksheets of Excel workbook. We can modify that worksheets keeping rest static and no need to run SAS code for each worksheet. Whereas, ODS Excel creates new Excel workbook after each execution of ODS Excel so we can't have static worksheet in our Excel and it is required to run SAS code for all worksheets.

CONCLUSION

We need not worry when someone needs customized output or SAS data in Excel workbook. ODS Excel is the best solution to export SAS data to Excel workbooks with multiple worksheets as needed. ODS Excel makes our output more reviewer friendly using options and suboptions.

REFERENCES

https://blogs.SAS®.com/content/SAS®_dummy/2014/08/29/experimenting-with-ods-excel-to-create-spreadsheets-from-SAS®/

https://blogs.SAS®.com/content/SAS®_dummy/2012/02/11/export-excel-methods/

https://blogs.SAS®.com/content/SAS®_training/2017/04/18/control-name-excel-sheets-created/

http://www.scsug.org/wp-content/uploads/2017/10/Chevell_SCSUG.pdf

ACKNOWLEDGMENTS

I would like to thank Raghu Kishor Koukuntla and all my peers who shared their experiences, read my drafts and gave me valuable and constructive feedback.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Author Name: Meghana Mahajani

Company: Cytel Statistical Software & Services Pvt. Ltd., Pune, India

T +91(20) 6709-0226

Time Zone: UTC+05:30

Email: meghana.mahajani@cytel.com

Brand and product names are trademarks of their respective companies

APPENDIX

SAS Code 2 – SAS Code that report SAS data ADSL in multiple Excel worksheets for four different Age categories

```
ods excel file = "<Path>\filename.xlsx";
proc report data=ADSL;
  by agecat;
  column subject treatment gender Height_Cm Weight_Kg bmi score;
  define subject / "Subject ID";
  define treatment / "Treatment";
  define gender / "Gender";
  define height_cm / "Height (in cm)";
  define weight_kg / "Weight (in kg)";
  define bmi / "BMI";
  define score / "Score";
run;
ods excel close;
```

SAS Code 3 - SAS code which write and customized ADSL in multiple Excel worksheets for Age categories

```
proc format; /*Highlight 'Score' column by red if score is 0-1*/
  value minscore
    0-1='Red';
run;

title1 j=1 "Study ID: AABB";
title2 j=1 "Listing for subjects with baseline information";
footnote1 j=1 "Note: Subjects listed in the output who received at least one Study
treatment.";

ods excel file = "<Path>\filename.xlsx"
  options(start_at = "2,2" /*Options */
    embedded_titles = "on"
    embed_footnotes_once = "on"
    autofilter = "1-7"
    frozen_headers = "on"
    frozen_rowheaders = "2"
    absolute_column_width = "15,16,13,10,15,15,14,14"
    row_heights = "30,20,50,20,20,25,5"
    sheet_name = "#byvall"
    tab_color = "yellow")
  style=Listing;

title3 j=1 "Age Category: #byvall";
footnote2 j=1 "Note: Subjects listed in the output whose baseline age is #byvall";

options nobyline;
proc report data=ADSL style(column)=[just=c]
  style(header)=[color=white backgroundcolor=grey] ;
  by agecat;
  column subject treatment gender ("Baseline Information" Height_Cm Weight_Kg bmi
score );
  define subject / "Subject ID" ;
  define treatment / "Treatment" ;
  define gender / "Gender" ;
  define height_cm / "Subject Height (in cm)" style(header)=[tagattr="wrap:yes"];
  define weight_kg / "Subject Weight (in kg)" style(header)=[tagattr="wrap:yes"];
  define bmi / "Subject BMI" ;
  define score / "Score" style(column)=[background=minscore.] ;
run;
ods excel close;
```

SAS Code 4 - SAS code which write and customized ADSL in multiple Excel worksheets for Age categories and Summary for baseline Characteristic

```
proc format; /*Highlight 'Score' column by red if score is 0-1*/
  value minscore
    0-1='Red';
run;

ods excel file = "<Path>\filename.xlsx";

title1 j=1 "Study ID: AABB";
title2 j=1 "Listing for subjects with baseline information";
footnote1 j=1 "Note: Subjects listed in the output who received at least one Study
treatment.";

ods excel options( embedded_titles = "on" /*Options for first report*/
  embed_footnotes_once = "on"
  autofilter = "1-7"
  tab_color = "yellow"
  frozen_headers = "on"
  frozen_rowheaders = "2"
  start_at = "2,2"
  absolute_column_width = "15,16,13,10,15,15,14,14"
  sheet_name = "#byvall1"
  row_heights = "30,20,50,20,20,25,5" )
  Style = Listing;

title3 j=1 'Age Category: #byvall1';
footnote2 j=1 'Note: Subjects listed in the output whose baseline age is #byvall1';

options nobyline; /*first reporting part*/
proc report data=ADSL style(column)=[just=c]
  style(header)=[color=white backgroundcolor=grey];
  by agecat;
  column subject treatment gender ("Baseline Information" Height_Cm Weight_Kg bmi
  score );
  define subject / "Subject ID" ;
  define treatment / "Treatment" ;
  define gender / "Gender" ;
  define height_cm / "Subject Height (in cm)" style(header)=[tagattr="wrap:yes"];
  define weight_kg / "Subject Weight (in kg)" style(header)=[tagattr="wrap:yes"];
  define bmi / "Subject BMI" ;
  define score / "Score" style(column)=[background=minscore.];
run;

title;
footnote;

title1 j = 1 "Summary table for Baseline Characteristic";
ods excel options(start_at = "1,2"
  embedded_titles = "on"
  autofilter = "all"
  frozen_headers = "on"
  frozen_rowheaders = "1"
  absolute_column_width = "15,12,10,10,15,15,14,14"
  row_heights = "30,20,50,20,20,25,5"
  sheet_name = "summary"
  tab_color = "green"
  )
  Style = Listing;
```

```
proc report data=all_stat style(header)=[color=white backgroundcolor=grey]
    style(column)=[just=c]
        nowd Headline headskip split="^" missing;

column agecatn agecat treatment gender trt_count sex_count height_sum weight_sum
    bmi_sum score_sum;

define agecatn / noprint order;
define agecat / "Age ^ category";
define Treatment / "Treatment";
define gender / "Sex" ;
define trt_count / "Treatment ^ N";
define sex_count / "Sex ^ N" ;
define height_sum / "Height ^ Mean(SD)";
define weight_sum / "Weight ^ Mean(SD)";
define bmi_sum / "BMI ^ Mean(SD)";
define score_sum / "Score ^ Mean(SD)";
compute after agecatn;
    line "";
endcomp;

run;
ods excel close;
```